
1 Corrections required

1.1 fl_TC_22_5_6_AttachAccept
	Template name
	fl_TC_22_5_6_AttachAccept

	Reason for change
	In the current TTCN implementation of this testcase, the function fl_TC_22_5_6_AttachAccept is called at multiple steps. However, within this function, In the current TTCN implementation, when the ESM message is not a dummy message, the value for IE TAIlist is being hardcoded to cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}) in the ATTACH accept message.
This needs correction.

	Summary of change
	Instead of passing cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}) in the ATTACH accept message, passed v_TaiList.

Please see below.

	TTCN module
	NBIOT_NAS_Attach.ttcn

	MCC160 Comment
	


Before Change:

	function fl_TC_22_5_6_AttachAccept(NBIOT_CellId_Type p_CellId,
                                     NAS_MSG_Indication_Type p_NAS_Ind,
                                     template (omit) TrackingAreaIdList p_TaiList := omit,
                                     template (omit) MobileIdentity p_Guti := omit,
                                     SecurityHeaderType p_SecurityHeader := tsc_SHT_IntegrityProtected_Ciphered,
                                     boolean p_MacErr := false,
                                     IOT_STATE_Type p_IOT_State := CONTROL_PLANE) runs on NBIOT_PTC
  {
    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);
    var template (value) NAS_MSG_Request_Type v_NAS_MSG_Request;
    var template (value) MobileIdentity v_Guti;
    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);
    var NAS_UL_Message_Type v_ESMMessage := p_NAS_Ind.Pdu.PiggybackedPduList[0].Msg;
    var EUTRA_ASN1_TrackingAreaCode_Type v_TAC := f_NBIOT_CellInfo_GetTAC(p_CellId);
    var EPS_BearerIdentity v_EpsDefaultBearerId := tsc_EpsDefaultBearerId;
    var template (value) EPS_NetworkFeatureSupport v_NtwkFeatSupport;
    var boolean v_IPv4AllocationViaNasFlag := true;
    var template (omit) ExtdProtocolConfigOptions v_PcoToUE := omit;
    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;
    var NBIOT_PDN_Type v_PDN;
    var template (value) PDN_Address v_PDN_Address;
    var template (omit) AdditionalUpdateResult v_AdditionalUpdateResult;
    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;
    var template (value) TrackingAreaIdList v_TaiList;
    if (isvalue(p_TaiList)) {
      v_TaiList := valueof(p_TaiList);
    } else {
      v_TaiList := cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)});
    }
    if (isvalue(p_Guti)) {
      v_Guti := valueof(p_Guti);
    } else {
      v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);
    }
    v_NtwkFeatSupport := f_GetNBIOTNetworkFeatureSupport(p_IOT_State);
    v_AdditionalUpdateResult  := f_GetNBIOTAdditionalUpdateResult(p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType);
    if (f_NBIOT_MobileInfo_GetNoOfPDNsConfigured() > 0) { // PDN Requested
      v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn (0);  // This is registration, so this is the first PDN
      v_PDN.pti := v_ESMMessage.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // @sic R5s180057 sic@
      v_PcoFromUE := v_PDN.protocolConfigurationOptions; // @sic R5s170597 sic@
      if (isvalue(v_PcoFromUE)) { // @sic R5s170597 sic@
        v_IPv4AllocationViaNasFlag := f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE);
      }
      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDN.assignedPdnType, v_IPv4AllocationViaNasFlag, -, pdnTypeAsSupportedByUE); // @sic R5s170597 sic@
      v_PcoToUE := f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE); // @sic R5s170597 sic@
    }
    if (ischosen(v_ESMMessage.eSM_DUMMY_MESSAGE)) { //@sic R5s170770 sic@
      v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(p_SecurityHeader,
                                                          cs_CIOT_ATTACH_ACCEPT(v_TaiList,
                                                                                v_Guti,
                                                                                v_NtwkFeatSupport,
                                                                                v_AdditionalUpdateResult),
                                                          cs_ESM_DUMMY_MESSAGE(v_ESMMessage.eSM_DUMMY_MESSAGE.procedureTransactionIdentifier));
    } else {
      v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(p_SecurityHeader,
                                                          cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                                v_Guti,
                                                                                v_NtwkFeatSupport,
                                                                                v_AdditionalUpdateResult),
                                                          cs_CIOT_ActivateDefEpsBearerContextRequest(v_EpsDefaultBearerId,
                                                                                                     v_PDN.pti,
                                                                                                     v_PDN.accessPointName,
                                                                                                     v_PDN_Address,
                                                                                                     v_PcoToUE,
                                                                                                     -,
                                                                                                     -,
                                                                                                     -,
                                                                                                     v_HeaderCompressConfigTX,
                                                                                                     cs_ControlPlaneInd,
                                                                                                     cs_ServingPLMNRateControl('FFFF'O)));
    }
    if (p_MacErr) {
      v_NAS_MSG_Request.SecurityProtection.ForceMacError := true; // force an integrity MAC error
    }
    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   v_NAS_MSG_Request));
  }


After Change:

	function fl_TC_22_5_6_AttachAccept(NBIOT_CellId_Type p_CellId,
                                     NAS_MSG_Indication_Type p_NAS_Ind,
                                     template (omit) TrackingAreaIdList p_TaiList := omit,
                                     template (omit) MobileIdentity p_Guti := omit,
                                     SecurityHeaderType p_SecurityHeader := tsc_SHT_IntegrityProtected_Ciphered,
                                     boolean p_MacErr := false,
                                     IOT_STATE_Type p_IOT_State := CONTROL_PLANE) runs on NBIOT_PTC
  {
    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);
    var template (value) NAS_MSG_Request_Type v_NAS_MSG_Request;
    var template (value) MobileIdentity v_Guti;
    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);
    var NAS_UL_Message_Type v_ESMMessage := p_NAS_Ind.Pdu.PiggybackedPduList[0].Msg;
    var EUTRA_ASN1_TrackingAreaCode_Type v_TAC := f_NBIOT_CellInfo_GetTAC(p_CellId);
    var EPS_BearerIdentity v_EpsDefaultBearerId := tsc_EpsDefaultBearerId;
    var template (value) EPS_NetworkFeatureSupport v_NtwkFeatSupport;
    var boolean v_IPv4AllocationViaNasFlag := true;
    var template (omit) ExtdProtocolConfigOptions v_PcoToUE := omit;
    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;
    var NBIOT_PDN_Type v_PDN;
    var template (value) PDN_Address v_PDN_Address;
    var template (omit) AdditionalUpdateResult v_AdditionalUpdateResult;
    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;
    var template (value) TrackingAreaIdList v_TaiList;
    if (isvalue(p_TaiList)) {
      v_TaiList := valueof(p_TaiList);
    } else {
      v_TaiList := cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)});
    }
    if (isvalue(p_Guti)) {
      v_Guti := valueof(p_Guti);
    } else {
      v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);
    }
    v_NtwkFeatSupport := f_GetNBIOTNetworkFeatureSupport(p_IOT_State);
    v_AdditionalUpdateResult  := f_GetNBIOTAdditionalUpdateResult(p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType);
    if (f_NBIOT_MobileInfo_GetNoOfPDNsConfigured() > 0) { // PDN Requested
      v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn (0);  // This is registration, so this is the first PDN
      v_PDN.pti := v_ESMMessage.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // @sic R5s180057 sic@
      v_PcoFromUE := v_PDN.protocolConfigurationOptions; // @sic R5s170597 sic@
      if (isvalue(v_PcoFromUE)) { // @sic R5s170597 sic@
        v_IPv4AllocationViaNasFlag := f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE);
      }
      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDN.assignedPdnType, v_IPv4AllocationViaNasFlag, -, pdnTypeAsSupportedByUE); // @sic R5s170597 sic@
      v_PcoToUE := f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE); // @sic R5s170597 sic@
    }
    if (ischosen(v_ESMMessage.eSM_DUMMY_MESSAGE)) { //@sic R5s170770 sic@
      v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(p_SecurityHeader,
                                                          cs_CIOT_ATTACH_ACCEPT(v_TaiList,
                                                                                v_Guti,
                                                                                v_NtwkFeatSupport,
                                                                                v_AdditionalUpdateResult),
                                                          cs_ESM_DUMMY_MESSAGE(v_ESMMessage.eSM_DUMMY_MESSAGE.procedureTransactionIdentifier));
    } else {
      v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(p_SecurityHeader,
                                                          cs_CIOT_ATTACH_ACCEPT(v_TaiList, //WA#22_5_6 cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                                v_Guti,
                                                                                v_NtwkFeatSupport,
                                                                                v_AdditionalUpdateResult),
                                                          cs_CIOT_ActivateDefEpsBearerContextRequest(v_EpsDefaultBearerId,
                                                                                                     v_PDN.pti,
                                                                                                     v_PDN.accessPointName,
                                                                                                     v_PDN_Address,
                                                                                                     v_PcoToUE,
                                                                                                     -,
                                                                                                     -,
                                                                                                     -,
                                                                                                     v_HeaderCompressConfigTX,
                                                                                                     cs_ControlPlaneInd,
                                                                                                     cs_ServingPLMNRateControl('FFFF'O)));
    }
    if (p_MacErr) {
      v_NAS_MSG_Request.SecurityProtection.ForceMacError := true; // force an integrity MAC error
    }
    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   v_NAS_MSG_Request));
  }


